STUDIECENTRUM VOOR ECONOMISCH EN SOCIAAL ONDERZOEK

VAKGROEP MACRO-ECONOMIE

A TSP-procedure
to test for the order of integration of a time series
by means of (A)DF, PP and KPSS tests

Charlotte Van Hooydonk (*)

report 95/329

October 1995

(*) The author wishes to thank without implicating Robert Van Straelen and Hugo Pauwels for helpful comments.

Universitaire Faculteiten St.-Ignatius
Prinsstraat 13 - B 2000 Antwerpen

D/1995/1169/16

Summary:

This paper proposes an easy-to-use TSP-subroutine to test for the order of integration of a time series by means of unit root and stationarity tests. Its main purpose is to offer a wider choice of tests and options than is commonly provided in statistical software packages.

More specifically, four of the procedure's main features are that (i) besides (A)DF tests, it also performs Phillips & Perron (PP) as well as KPSS stationarity tests, (ii) it offers a menu of alternative criteria by which to select the appropriate lag length (ADF tests) or number of autocovariances (PP and KPSS tests), (iii) it uses a new set of critical values for the KPSS tests and (iv) it takes account of the often ignored recommendation by Dickey & Pantula (1987) to test for the presence first of, say, three, then two, and finally one unit root.

After a brief description of the (A)DF, PP and KPSS tests, we discuss how the TSP subroutine works and what its output looks like. We also present the outcome of applying this procedure to a set of macroeconomic variables.

UFSIA - University of Antwerp, 13 Prinsstraat, B-2000 Antwerp, BELGIUM
Phone: ++32 3220 4068 or 4994, Fax: ++32 3220 4799 or 4026
E-mail: fte.vanhooydonk.c@alpha.ufsia.ac.be

1. Introduction

Over the last years, a variety of test procedures has been proposed to investigate the order of integration of a time series. The most popular statistics are to our mind unit root tests of the Dickey-Fuller family, viz. on the one hand (augmented) Dickey-Fuller tests (henceforth *(A)DF tests*) and on the other hand Phillips & Perron tests (*PP tests*).

Nowadays, most statistical packages enable the researcher to perform unit root tests of the (A)DF kind by issuing a trivial command. While the packages supply the user with a number of variants of these tests, it is often not clear which option(s) ought to be chosen. In this paper we therefore present a TSP-subroutine to test for unit roots via (A)DF or PP tests that seeks some of the appropriate options autonomously according to user-defined criteria. The routine also allows the researcher to perform tests of the null hypothesis of stationarity developed by Kwiatkowski e.a. (1992), i.e. *KPSS* tests.

The structure of the paper is as follows. First, we briefly review the concepts of "(covariance) stationarity", "order of integration" and "unit root" (section 2). We then describe the ordinary Dickey-Fuller (*DF*) test (and how it ought to be modified into an *ADF* or *PP* test in the presence of serial correlation) as well as the supplementary *KPSS* stationarity test (section 3).

In section 4 we propose a commented TSP-subroutine to test for the order of integration by means of (A)DF, PP as well as KPSS tests that (i) automatically determines the lag length (ADF tests) and the number of autocovariances (PP and KPSS tests), according to a user-defined criterion and (ii) automatically incorporates the suggestion of Dickey & Pantula (1987) to test for the presence of three, two and finally one unit root. This subroutine is invoked by giving a simple one-line command specifying the name of the variable to be examined as well as a limited number of option parameters.

The fifth section comments on the printed results of this TSP-procedure for some macroeconomic variables. The final section concludes.

2. Stationarity, order of integration and unit roots

A time series x_t is said to be integrated of order *d* (denoted by $x_t \sim I(d)$), if we have to difference it *d* times in order to get a "(covariance) stationary" series (Cuthbertson e.a., 1992, 130; Engle & Granger, 1987, 252).

A variable x_t is "(covariance) stationary" or "integrated of order zero" (denoted by $x_t \sim I(0)$), if

(i) $\forall t$: $E(x_t) = \mu$
 it has a constant, finite mean;

(ii) $\forall t$: $E[(x_t - \mu)^2] = var(x_t) = g_0$
 it has a constant, finite variance; and

(iii) $\forall t,k$: $E[(x_t - \mu).(x_{t+k} - \mu)] = cov(x_t,x_{t+k}) = g_k$
 its autocovariances (i.e. the covariances between any two observations) depend only upon the time distance between both observations (Judge e.a., 1988, 678).

According to this definition a stationary series has mean, variance and autocovariances that are independent of time. This implies that no fundamental breaks appear in the series's structure (Judge e.a., 1988, 679; Harvey, 1989, 49). A stationary series will indeed frequently take the value of its own (constant) mean and fluctuate around this mean within a constant band.

On the contrary, non-stationary series could show a mean that fluctuates through time, e.g. because of a trend. We then have to distinguish between two sources of trend behaviour and two ways of transforming the series into a stationary one.

If in

$$x_t = a + b.x_{t-1} + c.t + e_t \quad (1)$$

$|b| < 1$ and $c \neq 0$, we say that x_t is a "trend stationary" process because the fluctuations around the deterministic trend c.t are stationary ($|b| < 1$). Including a time trend in the regression will then suffice to make e_t stationary.

If, however, b=1 and c=0 in equation (1), we get:

$x_t = a + x_{t-1} + e_t$

or: $\Delta x_t = a + e_t$

The series x_t is then called a "random walk" with "drift parameter" a. The only way to remove the stochastic trend out of the series and to make it integrated of order zero is to take first differences: $x_t \sim I(1)$ and $\Delta x_t \sim I(0)$. The random walk with drift is indeed a "difference stationary" process (Stewart, 1991, 202-3, Hamilton, 1994, ch. 15, Harris, 1995, 18-20).

The economic meaning of a random walk process is that at least part of a time series's movements ("shocks" or "innovations") does not have a temporary but a permanent impact on the variable's level (Campbell & Perron, 1992, 46). We say that such variables contain a "unit root" (see also section 3).

3. (Augmented) Dickey-Fuller and Phillips-Perron tests for unit roots and Kwiatkowsi e.a. test for stationarity

In this section we briefly review the unit root tests proposed by Dickey & Fuller and by Phillips & Perron as well as the stationarity test of Kwiatkowski e.a. All these tests are included in the TSP-procedure (see section 4).

a. Dickey-Fuller (DF) tests

Investigating whether an AR(1)-process is stationary boils down to testing whether H_0: a=1 (or, equivalently, $x_t \sim I(1)$, i.e. x_t contains a unit root) can be rejected in favour of H_1: $|a| < 1$ ($x_t \sim I(0)$) in

$$x_t = a.x_{t-1} + e_t \qquad (2)$$

We call this test a "test for the presence of a unit root". This becomes clear when we rewrite the AR(1)-process as:

$(1 - a.L).x_t = e_t$ where $Lx_t = x_{t-1}$

Like any autoregressive process, the process is stationary if "all roots z_0 of the polynomial (1 - a.z) have a modulus $|z_0|>1$". In the trivial case of an AR(1)-process the only root is $z_0 = 1/a$. It is obvious that the process is stationary if $|1/a|>1$ or $|a|<1$, which is indeed the same as the alternative hypothesis H_1. The AR(1)-process is on the contrary non-stationary if there is a modulus $|z_0|=1$, i.e. a "unit root" (Judge e.a., 1988, 681).

The test whether H_0 can be rejected consists in comparing the t-value τ of the estimated parameter $\hat{a}$ with the critical values τ_{cr} tabulated by a.o. Fuller (1976) and MacKinnon (1991)[1]. We reject H_0 if $|\tau| > |\tau_{cr,\alpha}|$, where α is the significance level (Cromwell e.a., 1994, 13-4).

Often the alternative representation of (2) is found, viz.:

$$\Delta x_t = (a_1-1).x_{t-1} + e_t \tag{3}$$

which can be supplemented with a constant and/or trend term:

$$\Delta x_t = a_0 + (a_1-1).x_{t-1} + e_t \tag{4}$$

$$\Delta x_t = a_0 + (a_1-1).x_{t-1} + a_2.t + e_t \tag{5}$$

The tests whether H_0: $a_1 = 1$ is valid in (3), (4) and (5) are called *Dickey-Fuller (DF) tests*. As the critical values are obtained under the hypothesis that $a_0 = 0$ en $a_2 = 0$, the regression equations (4) and (5) are overspecified. The real nulls are in fact:

H_0: $a_1=1$ for (3)

H_0: $a_1=1$ and $a_0=0$ for (4)

H_0: $a_1=1$ and $a_0=a_2=0$ for (5)

Using the tabulated critical values implies thus that one accepts the zero-restrictions in the real nulls of (4) and (5) (Harris, 1995, 28-32)[2]. Perron (1988, 303-4) contends that (3) is

---

1 - We cannot use the usual Student-t-distribution because of non stationarity under H_0.
- The TSP-procedure uses MacKinnon's critical values.

2 It has been shown that, if the regression equations are well specified (viz. $a_0 \neq 0$ in (4) and $a_0 \neq 0$, $a_2 \neq 0$ in (5)), the t-values of $(\hat{a}_1-1) \sim N(0,1)$ under H_0: $a_1=1$. This means that the usual t-tables provide an adequate approximation for large samples. In small samples however the distribution depends heavily on the values of a_0 in (4) and a_2 in (5). The Dickey-Fuller tables might then after all yield better results in limited samples with small values of a_0 and a_2 (Stewart, 1991, 200-1; Schmidt & Phillips, 1992, 258;

adequate if one suspects that the series x_t is a random walk without drift with zero mean; (4) if one supposes that x_t is a random walk without drift with non-zero mean and (5) when x_t is assumed to be a random walk with drift. Hamilton (1994, 501) advises to "fit a specification that is a plausible description of the data under both the null hypothesis and the alternative". He suggests using (5) for series with an obvious trend (in order to test the null of a stochastic trend against the alternative of a deterministic trend) and (4) for series without a significant trend. This implies that the choice will be mostly between (4) and (5), making (3) hardly appropriate for most economic time series.

While these DF-tests remain valid in the presence of heteroscedasticity or anormality of the residuals, they are not robust in the presence of serial correlation (MacKinnon, 1991, 270; Hamilton, 1994, 502). Two modifications to the ordinary DF tests have been proposed to cope with this problem: ADF and PP tests.

b. Augmented Dickey-Fuller (ADF) tests

In order to cope with serial correlation one could perform so-called "augmented" Dickey-Fuller (ADF) tests (same critical values as DF tests)[3]:

$$\text{ADF(j)}: \Delta x_t = (b_0) + b_1.x_{t-1} + (b_2.t) + \Sigma_{i=1}{}^{j} c_i.\Delta x_{t-i} + e_t \qquad (6)$$

The proposed correction to the ordinary DF-test thus involves adding to the right hand side of the DF-specification as many lagged first differences of x_t as necessary to remove all serial correlation (MacKinnon, 1991, 270). The determination of the appropriate lag length *j* is not straightforward. Cautious selection is called for, as too few lags give rise to poor size properties, while too many lags might lead to low power (Harris, 1995, 34).

Harris, 1995, 31).

[3] H_0: $b_1=0$ posits that the sum of the AR(p+1)-coefficients equals 1, i.e. x_t contains a unit root (e.g. Harvey, 1990, 82; Cuthbertson e.a., 1992, 136). Indeed the AR(p+1)-process

$$x_t = a_1.x_{t-1} + \ldots + a_{p+1}.x_{t-p-1} + e_t$$

can be rewritten as

$$\Delta x_t = (\Sigma_{i=1}{}^{p+1} a_i - 1).x_{t-1} - \Sigma_{i=1}{}^{p} \Sigma_{j=i+1}{}^{p+1} a_j.\Delta x_{t-i}, \text{ or:}$$

$$\Delta x_t = b_1.x_{t-1} + \Sigma_{i=1}{}^{p} c_i.\Delta x_{t-i}$$

with: $b_1 = \Sigma_{i=1}{}^{p+1} a_i - 1$

$$c_i = \Sigma_{j=i+1}{}^{p+1} a_j$$

We distinguish between two alternative routes to determine the appropriate ADF-order j, viz.:

(1) starting from the ordinary DF test regression and adding an extra lagged difference whenever there is evidence of serial correlation, until all serial correlation is removed (see e.g. MacKinnon, 1991, 270), or

(2) starting from a general specification with a prespecified maximal value of j and then performing t- and F-tests to see whether the coefficient(s) of the last incuded lagged difference(s) can be dropped (see e.g. Hamilton, 1994, 530).

Other options are setting the lag length by

(3) applying one of Schwert's rules - Schwert (1989) suggests using a deeper lag length with growing sample size, according to the following formulae: j = integer $[4.(T/100)^{0.25}]$ or j = integer $[12.(T/100)^{0.25}]$.

(4) minimizing Akaike's Information Criterion (AIC) - the use of this criterion is cautioned against in Harris (1995, 36)

The TSP-procedure provides all options, although (4) only in an indirect way (see section 4). The user has to specify his choice by giving a value for the option parameter *adfway*.

c. Phillips-Perron (PP) tests

Phillips (1987) and Perron (1988) proposed a test for unit roots in the presence of serial correlation which comes down to modifying the value of the DF t-test statistic. Their modified test statistic is given by (Hamilton, 1994, 514):

$$Z = (g_0/\lambda^2)^{0.5}.t_{DF} - (1/2).(\lambda^2 - g_0).(1/\lambda).(T.se/s) \qquad (7)$$

with:
- g_j = j-th autocovariance of the OLS residuals = $(1/T).\Sigma_{t=j+1}{}^{T} e_t.e_{t-j}$
- g_0 = $(1/T).\Sigma_{t=1}{}^{T} e^2{}_t$
- λ^2 = (Newey-West) estimator of the variance of the sample mean of e_t = $g_0 + 2.\Sigma_{j=1}{}^{q}[1 - j/(q+1)].g_j$
- λ = $\sqrt{\lambda^2}$
- q = number of autocovariances deemed relevant
- t_{DF} = t-value of coefficient of x_{t-1} (= DF test value)
- T = number of observations
- se = estimated standard error of coefficient of x_{t-1} (i.e. denominator of t_{DF})

s^2 = estimated standard error of the regression
s = $\sqrt{s^2}$

Thus, after estimating the DF test regression, one has to save some additional parameters (viz. se and s^2), calculate autocovariances of the residuals g_j up to the order q, use all these parameters to estimate the variance of the residuals λ^2, and then calculate the value of the Z-statistic. These Z-statistics have the same distribution as the (A)DF test statistics, so once more the critical values of MacKinnon (1991) can be used (Hamilton, 1994, 514).

d. KPSS tests

As the power of all unit root tests developed so far, is not very high, it might be useful to supplement them with a test of the null hypothesis of (level or trend) stationarity. Recently, such a test has been developed by Kwiatkowski e.a. (1992). Their KPSS test statistic is:

$$N = (1/T^2).\Sigma_{t=1}{}^{T}S_t{}^2/\lambda^2 \qquad (8)$$

with: T : same as in PP tests (see equation 7)
λ^2 : same as in PP tests (see equation 7)
$S_t{}^2$ = $\Sigma_{i=1}{}^{t}e_i$
e_i = OLS residuals of regression of x on a constant (H_0: x is level stationary) or on a constant and a trend (H_0: x is trend stationary)

Appropriate critical values are supplied by Sephton (1995).

The TSP-procedure calculates the value of λ^2 for PP and KPSS tests in exactly the same way. It allows the user to determine q according to the same criteria as the order j of the augmented Dickey-Fuller tests (by means of the option parameter *qway*, see section 4).

On comparing the results of unit root tests with those of stationarity tests any of the following four cases may apply[4]:

(i) the null of a unit root is not rejected, while the null of stationarity is. This is to be

[4] Cfr. Harris & Inder (1994, 134) in the case of KPSS cointegration tests.

regarded as strong evidence in favour of a unit root.

(ii) the null of a unit root is rejected, while the null of stationarity is not. This leads to the conclusion of stationarity.

(iii) both nulls are rejected: possibly, a type I error has occurred.

(iv) both nulls are not rejected: the tests lack power.

As contradictions between the outcome of various unit root tests and stationarity tests may occur and as the power and size properties of all these tests are rather poor, performing them should have as its main purpose to investigate whether the data in a time series *might* have been generated by a unit root process or not. Thus, when we say that variable such and such "contains" a unit root, we mean that the series can be reasonably approximated by a unit root process (see also Harris, 1995, 47).

Two final remarks are in order. First, to avoid wrong conclusions, one ought to test for a descending order of integration. In other words, one should first test for the presence of d unit roots, then for d-1, d-2,... and at last for the presence of 1 unit root (Dickey & Pantula, 1987). The TSP-procedure presented below indeed automatically tests whether the series under investigation is integrated of order three, two or one.

Second, it is well-known that the presence of a break in a (trend) stationary time series might lead to rejection of the null of a unit root. The TSP-procedure does not take into account these effects - the interested reader is referred to e.g. Harris (1995, 40-1).

4. A TSP-procedure to test for the order of integration of a time series

The TSP-procedure presented in this section allows the researcher to perform unit root and stationarity tests along the lines laid down in section 3. In the next subsections we will describe (a) how the procedure works and (b) what its output looks like. The final subsection (c) contains the commented TSP-command file.

a. How the procedure works

The procedure only works in full batch mode. This means that the user has to prepare a

TSP-command file, say "c:\tsp\myfile.tsp", including the data, the procedure and a one-line command. Setting TSP to work is done by typing "tsp myfile.tsp" after the c:\tsp\ prompt. The results file will then automatically be stored under the name "c:\tsp\myfile.out". We will now comment on the various ingredients of the user's command file.

First, the data can be loaded either directly or by reference to a database. In either case, the time series ought to be loaded *as if they contained cross-sectional or survey data,* that is, with the TSP-command "freq n", with the sample from 1 to T, the number of observations.

Second, the whole of the TSP-procedure in subsection c is to be loaded into the user's command file[5].

Third, the one-line command setting TSP to work specifies the name of the time series to be examined, the number of observations and the value of 6 option parameters, or, in general terms:

inttest z, numobs, sign, opt, adfway, qway, corrorde, maxorder;

with:

inttest	the name of the TSP-procedure (fixed)
z	the name of the time series under investigation
numobs	the number of observations in the sample
sign	the significance level => <u>limited</u> to 0.01, 0.05 or 0.10
opt	the kind of unit root/stationarity test(s) to be performed => <u>limited</u> to 1, 2, 3, 4, 5 or 6
adfway	the way in which the lag length of the (A)DF tests (*j*) ought to be performed => <u>limited</u> to 1, 2, 3.1, 3.2 or 4
qway	the way in which the number of autocovariances (*q*) is to be determined => <u>limited</u> to 1, 2, 3.0, 3.1, 3.2 or 4
corrorde	the highest order of serial correlation to test for (ADF tests) => <u>limited</u> to 2, 4, 8 or 12
maxorder	the maximum order of the ADF tests, needed when performing (A)DF tests.

The TSP-procedure will then perform tests for the presence of three, two and one unit

[5] The procedure is available from the author at no cost. It suffices to send a formatted 3.5" diskette and a self-addressed envelope to the author's address.

roots in the variable *z* specified by the user. In other words, the procedure performs (A)DF, PP and/or KPSS tests on the series of second and first differences as well as on the series in levels (in this order), using the significance level *sign* throughout[6].

The researcher has to determine the kind of test(s) to be carried out by specifying the value of the parameter *opt*, viz.

opt = 1 => (A)DF tests
opt = 2 => PP tests
opt = 3 => KPSS tests
opt = 4 => ADF as well as PP tests (in this order)
opt = 5 => ADF as well as KPSS tests (in this order)
opt = 6 => ADF, PP and KPSS tests (in this order)

Three variants of each unit root test are carried out, viz. (1) no constant, no trend, (2) constant, no trend and (3) constant and trend (see equations (3), (4) and (5) respectively), while for the KPSS stationarity test the N statistics are calculated for the level stationary as well as trend stationary cases.

The other option parameters (*adfway*, *qway*, *corrorde* and *maxorder*) relate to the way in which these tests are carried out.

First, *adfway*, *corrorde* and *maxorder* specify how the ADF-lag length *j* is set:

If *adfway* = 1, the procedure starts with the simple DF-test regression and checks for residual correlation by means of Breusch-Godfrey LM-tests up to the order *corrorde*[7]. If there is evidence of serial correlation, the procedure raises the order *j* of the (A)DF-test by one and checks again for the presence of serial correlation. Only when no correlation is found, the (A)DF test is performed and the results printed. The user has to specify a maximum ADF-order *maxorder*. If *j* equals *maxorder* and still no ADF-test can be performed (due to stubborn serial correlation), no further tests are carried out and the order of integration is set to 10000.

If *adfway* is set to 2, the procedure starts by estimating the test regression for the maximum ADF-order *maxorder* and then tests the significance of the coefficient of the last lagged difference (i.e. $\Delta x_{t-maxorder}$) by means of a t-test. If this regressor can be dropped, an F-test on the joint significance of the parameters of

---

[6] The procedure uses *sign* in three instances: (1) when determining whether the Breusch-Godfrey LM-tests for serial correlation are significant (ADF tests when adfway=1), (2) when determining whether the t-tests and F-tests are significant (ADF tests when adfway=2) and (3) when calculating the appropriate critical values for the actual (A)DF, PP and KPSS tests.

[7] See e.g. Stewart (1991, 168-70) for a description of the LM-tests.

the lagged differences with the two highest lag values is performed, and so on, until no further regressors can be eliminated. Each time these zero-restrictions are unrejected, the order j of the (A)DF-test decreases by one. Only if no more coefficients can be dropped, the (A)DF(j)-test is carried out and the outcome is printed.

Alternatively, one might fix the ADF-order j according to Schwert's rules. If *adfway* equals 3.1, the formula j = integer $[4.(T/100)^{0.25}]$ is used; if it is 3.2, j = integer $[12.(T/100)^{0.25}]$.

Finally, setting *adfway* equal to 4 generates an overview of (A)DF(j) test results for different values of j (from 0 to 12), including AIC-values.

Second, *qway* determines how q, the number of autocovariances "deemed relevant", is set when calculating the PP and KPSS test statistics:

If *qway* equals 1, q is assumed to take the q value of user-defined *corrorde*.

If ADF type tests with *adfway* equal to 1 have been performed already, setting *qway* equal to 2 results in q taking the value of the actual order j of the performed ADF tests.

One might alternatively follow Schwert's simple rules to determine q, e.g. $q = 0$ (*qway* = 3.0) or q = integer$[4.(T/100)^{0.25}]$ (*qway* = 3.1) or q = integer $[12.(T/100)^{0.25}]$ (*qway* = 3.2).

A general overview of PP and KPSS test results with q ranging from 0 to 12 is given when *qway* is 4.

It is important to bear in mind that all option parameters *must* be specified, even if their value is irrelevant for the chosen test(s) (e.g. "qway" in the case of ADF tests).

b. Output of the procedure

First, TSP prints the value of *opt* and of *sign*.

Then, the procedure prints the outcome of the selected test(s), first for the series of double first differences of the variable under examination, then for the series of first differences and finally for the series in levels itself. For each variable, the outcome is thus presented in three blocks. Each of these is preceded by a print-out of the value of *nudiff* (the number of first differences taken), which obviously takes the value of 2, 1 and 0 respectively.

In each block the form of the output depends on the kind of test(s) selected. Two general remarks are in order. First, throughout the procedure, names ending in 1, 2 or 3, refer to the three variants of the unit root tests, viz. 1 for the "no constant, no trend" case, 2 for the "constant, no trend" case and 3 for the "constant and trend" case or to the two variants of the stationarity tests, viz. 2 for "level stationarity" and 3 for "trend stationarity". In the next paragraphs we replace 1, 2 and 3 by a dot "." for the sake of generality. Second, an ADF order of 0 means that the ordinary, i.e. non-augmented DF test of subsection 3.a is performed.

In the case of (A)DF tests, TSP prints:

J or J.	the order of the ADF test (J when adfway = 1; J. when adfway = 2, 3.1 or 3.2)
ADFT.	the t-test value
CR.	the appropriate MacKinnon (1991) critical value
INTORDE.	the order of integration (0 or 1) obtained by comparing ADFT. and CR.

If *adfway* equals 4, these results are printed for (A)DF-orders ranging from 0 to 12. For each of the three variants a (13*7) matrix "results." is printed, with the number of rows corresponding to the different j-values. The first column gives the j-values, the second the t test values, the third the critical values, the fourth the orders of integration, the fifth the values of Akaike's Information Criterion, the sixth the minimum of all tail probabilities attached to the Breusch-Godfrey LM-tests for serial correlation and the last the t-value attached to the j-period-lagged regressor.

In the case of PP tests, the TSP-output consists of:

Q	the number of autocovariances in the estimator of the error variance
Z.	the value of the Z-statistic
CR.	the appropriate MacKinnon (1991) critical value
INTORDE.	the order of integration (0 or 1) obtained by comparing Z. and CR.

If *qway* is set to 4, these results are stored for 13 values of q, so that for each of the three variants a (13*4) matrix "results." is printed. The number of rows correspond to the different q-values; in the first column these q-values are printed, in the second the z statistics, in the third the critical values and in the last the conclusions w.r.t. the order of integration.

Finally, in the case of KPSS tests, the TSP results are:

Q	the number of autocovariances in the estimator of the error variance
N.	the value of the N-statistic
CRKPSS.	the appropriate Sephton (1995) critical value
INTORDE.	the order of integration (0 or 1) obtained by comparing N. and CRKPSS.

If *qway* equals 4, these results are given for 13 values of q. In this case a (13*4) matrix "results." is printed for both variants of the KPSS tests. The first column gives the different q-values, the second the n statistics, the third the critical values and the last the orders of integration.

An example of the output of the TSP-procedure (in table format) is presented in appendix B (see section 5 for a description of the variables and for the TSP-procedure options chosen).

c. Commented TSP-procedure

Researchers using this TSP-procedure in this or in a modified form, are kindly requested to cite this working paper as their source.

```
proc inttest z,numobs,sign,opt,adfway,qway,corrorde,maxorder;
    regopt (lmlags=corrorde,noprint,pvcalc) all;
    set m=maxorder; supres @smpl;
    smpl 1 numobs; trend time;
    smpl 2 numobs;genr dz=z-z(-1);
    smpl 3 numobs;genr ddz=dz-dz(-1);
    smpl 4 numobs;genr dddz=ddz-ddz(-1);
    print opt, sign; set nudiff=2;
     dot ddz dz z;
      print nudiff; set start=nudiff+2; set startf=start-1;
      if opt=1 .or. opt=4 .or. opt=5 .or. opt =6; then; adftest d.,.,qadf0,qadf1,qadf2;
      if opt=2 .or. opt=4 .or. opt=6; then;
         do; smpl start numobs; philtest d.,.,qadf0,qadf1,qadf2; enddo;
      if opt=3 .or. opt=5 .or. opt=6; then;
         do; smpl startf numobs; kpsstest d.,.,qadf0,qadf1,qadf2; enddo;
      set nudiff=nudiff-1;
     enddot;
endproc;
```

This block generates a time trend and the series of first, second and third differences of the variable to be examined. It also tells TSP to examine the order of integration, first, of the series of double first differences ("nudiff"=2), then of the series of first differences ("nudiff"=1) and finally of the variable in levels ("nudiff"=0). The kind of unit root or stationary test(s) to be performed is determined by the value of the parameter "opt".

```
proc adftest dy,y,qadf0,qadf1,qadf2;
 if adfway=1 .or. adfway=4; then; do;
   set j=0; set h=0; set ok1=0;set ok2=0;set ok3=0;
   smpl start numobs;
   olsq dy y(-1); set nob=@nob; set aic1j0=@AIC; set tv1j0=0; minlm min1,adft1;
   olsq dy y(-1) c; set aic2j0=@AIC; set tv2j0=0; minlm min2,adft2;
   olsq dy y(-1) c time; set aic3j0=@AIC; set tv3j0=0; minlm min3,adft3;
   if adfway=1; then; test1;
   if adfway=4; then; do;
```

```
    dot 1 2 3; set nob.=nob; set min.j0=min.; set t.j0=adft.; enddot;
    critv cr1j0 cr2j0 cr3j0; enddo;
  if adfway=1; then; do;
   do j=1 to m;
    if h<3; then; do; set e=-j; set startb=start+j; smpl startb numobs;
      olsq dy y(-1) dy(-1)-dy(e);set nob=@nob; minlm min1,adft1;
      olsq dy y(-1) dy(-1)-dy(e) c; minlm min2,adft2;
      olsq dy y(-1) dy(-1)-dy(e) c time; minlm min3,adft3; test1; enddo;
    enddo;
  enddo;
  if adfway=4; then; do;
   set e=-1; set j0=0; set w=0; set v=2;
   dot 1 2 3 4 5 6 7 8 9 10 11 12;
      set startc=start-e; smpl startc numobs;
      olsq dy y(-1) dy(-1)-dy(e);set nob1=@nob; set aic1j.=@AIC; set tv1j.=@t(v); minlm min1j.,t1j.;
      olsq dy y(-1) dy(-1)-dy(e) c; set nob2=@nob; set aic2j.=@AIC; set tv2j.=@t(v); minlm min2j.,t2j.;
      olsq dy y(-1) dy(-1)-dy(e) c time; set nob3=@nob;set aic3j.=@AIC;set tv3j.=@t(v);minlm min3j.,t3j.;
      critv cr1j. cr2j. cr3j.;
      set e=e-1; set w=w+1; set j.=w; set v=v+1;
   enddot;
   dot 1 2 3;
      dot 0 1 2 3 4 5 6 7 8 9 10 11 12;
        if t.j.>cr.j.; then; set int.j.=1; else; set int.j.=0;
      enddot;
      mmake vect j0-j12,t.j0-t.j12,cr.j0-cr.j12,int.j0-int.j12,aic.j0-aic.j12,min.j0-min.j12,tv.j0-tv.j12;
      mform(type=general,nrow=13,ncol=7) results.=vect;
      print results.;
   enddot;
  enddo;
  if adfway=1; then; do;
   dot 1 2 3;
     if ok.=0; then; do;
        set intorde.=10000; print m intorde.;
        if nudiff=2;then;set qadf2=10000;
        if nudiff=1;then;set qadf1=10000;
        if nudiff=0;then;set qadf0=10000; enddo;
   enddot;
  enddo;
enddo;
if adfway=2; then; do;
 set k=m+1;
 dot 1 2 3;
  set startd=start+m; smpl startd numobs;
  set j.=10000; set e=-m;
  list varl1= y(-1) dy(-1)-dy(e);list varl2= varl1 c;list varl3= varl1 c time;
  olsq dy varl.;
  set ttest=%t(k); set sseu=@ssr; set nob.=@nob; set adft.=@t(1);
  set numdf1=nob.-1-m; set numdf2=nob.-2-m; set numdf3=nob.-3-m;
  if ttest<sign; then; set j.=m; else; do;
    set e=-m+1;
    list varl1= y(-1) dy(-1)-dy(e); list varl2= varl1 c; list varl3= varl1 c time;
    set starte=start-e; smpl starte numobs;
```

```
      olsq dy varl.;
      set nob.=@nob; set adft.=@t(1); set pnob.=nob.; set padft.=adft.;
      do i=3 to m;
        if j.=10000; then; do;
          set e =-m-1+i;
          list varl1= y(-1) dy(-1)-dy(e); list varl2= varl1 c; list varl3= varl1 c time;
          set starte=start-e; smpl starte numobs;
          olsq dy varl.;
          set sser=@ssr; set nob.=@nob; set adft.=@t(1); set numrest=i-1;
          set ftest=((sser-sseu)/(sseu*numrest))*numdf.;
          cdf(f,df1=numrest,df2=numdf.) ftest fkr;
          if fkr<sign; then; do;
            set j.=k-i+1; set nob.=pnob.; set adft.=padft.; enddo;
            else; do; set pnob.=nob.; set padft.=adft.; enddo;
        enddo;
      enddo;
      if j.=10000; then; do;
        list varl1= y(-1); list varl2= varl1 c; list varl3= varl1 c time;
        smpl start numobs;
        olsq dy varl.;
        set sser=@ssr; set nob.=@nob; set adft.=@t(1);
        set ftest=((sser-sseu)/(sseu*m))*numdf.;
        cdf(f,df1=m,df2=numdf.) ftest fkr;
        if fkr<sign; then; do;
          set j.=1; set nob.=pnob.; set adft.=padft.; enddo;
          else; set j.=0;
      enddo;
    enddo;
  enddot;
  test2;
 enddo;
 if adfway = 3.1 .or. adfway = 3.2; then; do;
   if adfway = 3.1; then; set ja=int(4*((numobs/100)**0.25));
   if adfway = 3.2; then; set ja=int(12*((numobs/100)**0.25));
   if ja < 1; then; do; smpl start numobs;
     olsq dy y(-1); set nob=@nob; set adft1=@t(1);
     olsq dy y(-1) c; set adft2=@t(1);
     olsq dy y(-1) c time; set adft3=@t(1);
   enddo; else; do;
   set e=-ja; set starth=start-e; smpl starth numobs;
     olsq dy y(-1) dy(-1)-dy(e); set nob=@nob; set adft1=@t(1);
     olsq dy y(-1) dy(-1)-dy(e) c; set adft2=@t(1);
     olsq dy y(-1) dy(-1)-dy(e) c time; set adft3=@t(1);
   enddo;
   set nob1=nob; set nob2=nob; set nob3=nob; set j1=ja; set j2=ja; set j3=ja;
   test2;
 enddo;
endproc;
```

This subprocedure carries out ADF tests.

If "adfway" equals 1, it starts from the ordinary DF test regression, checks for serial correlation and

carries out the (A)DF test only in the absence of serial correlation (see subprocedures "minlm" and "test1"). The procedure continues to try to perform (A)DF tests until an (A)DF test can be performed for all three variants of the same order "j" (note: the three variants are described above, viz.: (1) no constant, no trend; (2) constant, no trend and (3) constant and trend); when this happens the value of the control parameter "h" is set to three (and the (A)DF-order is saved in case the user wants to perform PP or KPSS tests with the value of "qway" set to 2).
If, for a given variant, after including up to "maxorder" lags, there is still serial correlation (so that no (A)DF test could be performed for this variant), the order of integration "intorde" is set to 10000. In this case the procedure prints the maximal order of the (A)DF tests "m" and the order of integration "intorde" (set to 10000).

If adfway is set to 2, the subprocedure starts from a general ADF(maxorder) test regression and performs a t-test to see whether the last lag can be excluded. If this is the case, an F-test is carried out to determine whether the last two lags can be dropped, and so on, until no further lags can be dropped and the appropriate order of the (A)DF test "j" is known at last. The subprocedure "test2" performs the actual (A)DF tests.

If adfway equals 3.1 or 3.2, the ADF-order is set according to Schwert's rules, the test is executed and the results printed.

*If adfway equals 4, the procedure performs (A)DF tests for all orders until 12. For each of the three variants a (13*7) matrix is printed containing the results: the orders of the test (j ranging from 0 to 12), the t test values, the critical values, the orders of integration, the values of AIC, the minimum of all tail probabilities attached to the Breusch-Godfrey LM-tests for serial correlation (see subprocedure "mimlm") as well as the t-value of the j-period-lagged regressor.*

```
proc minlm d,adft;
  set adft=@t(1);
  if corrorde=2;then;
    do;dot 1 2; set f.=%lmar.; enddot; mmake lmarser f1-f2;enddo;
  if corrorde=4;then;
    do;dot 1 2 3 4; set f.=%lmar.; enddot; mmake lmarser f1-f4;enddo;
  if corrorde=8;then;
    do;dot 1 2 3 4 5 6 7 8;set f.=%lmar.; enddot; mmake lmarser f1-f8;enddo;
  if corrorde=12;then;
    do;dot 1 2 3 4 5 6 7 8 9 10 11 12;set f.=%lmar.;enddot;mmake lmarser f1-f12;enddo;
  mat d=min(lmarser);
endproc;
```

This subprocedure is part of the (A)DF test procedure and is used when "adfway" is 1 or 4. It saves the value of the t-statistic of each test regression ("adft") and checks for serial correlation up to the order "corrorde" specified by the user. It does so by seeking the minimum of all tail probabilities attached to the Breusch-Godfrey LM-tests for serial correlation (the number of calculated LM-tests is "corrorde"), which is called "d" and later on "min.". Only when this "min." is insignificant, the (A)DF test is performed (see subprocedure "test1").

```
proc test1;
  set h1=0; set h2=0; set h3=0;
  set nob1=nob;set nob2=nob;set nob3=nob;
  critv cr1 cr2 cr3;
  dot 1 2 3;
   if min.>sign; then; do;
```

```
        set h.=1; set ok.=1;
        if adft.>cr.; then; set intorde.=1; else; set intorde.=0;
        print j adft. cr. intorde.; enddo;
    enddot;
    set h=h1+h2+h3;
    if h=3; then; do;
      if nudiff=2; then; set qadf2=j;
      if nudiff=1; then; set qadf1=j;
      if nudiff=0; then; set qadf0=j; enddo;
endproc;
```

This set of commands is part of the (A)DF tests subprocedure when "adfway" = 1. It calculates critical values (subprocedure "critv") and performs (A)DF tests only when the smallest value of the tail probabilities "d" is insignificant, i.e. when there is no evidence of serial correlation (see subprocedure "minlm"). If the t-value "adft" exceeds the critical value "cr", then the variable has an order of integration "intorde" of 1, else "intorde" is zero. The procedure prints the order of the (A)DF test "j", the t-value "adft", the critical value "cr" and the order of integration "intorde". The control parameter "h" is three only if an (A)DF test of the same order could be performed for all three variants. If "h" indeed equals 3, the subprocedure for (A)DF tests comes to an end (see subprocedure "adftest"). "qadf." is needed in case the user wants to perform PP or KPSS tests when "qway" = 2.

```
proc critv critv1 critv2 critv3;
 dot 1 2 3;
  if sign=0.01; then; do;
     set critv1=-2.5658 -1.960/nob. -10.04/(nob.**2);
     set critv2=-3.4335 -5.999/nob. -29.25/(nob.**2);
     set critv3=-3.9638 -8.353/nob. -47.44/(nob.**2); enddo;
  if sign=0.05; then; do;
     set critv1=-1.9393 -0.398/nob.;
     set critv2=-2.8621 -2.738/nob. - 8.36/(nob.**2);
     set critv3=-3.4126 -4.039/nob. -17.83/(nob.**2);enddo;
  if sign=0.10; then; do;
     set critv1=-1.6156 -0.181/nob.;
     set critv2=-2.5671 -1.438/nob. - 4.48/(nob.**2);
     set critv3=-3.1279 -2.418/nob. - 7.58/(nob.**2);enddo;
 enddot;
endproc;
```

This subprocedure calculates critical values given by MacKinnon (1991, 275). It is used when performing (A)DF as well as PP unit root tests.

```
proc test2;
  critv cr1 cr2 cr3;
  dot 1 2 3;
    if adft.>cr.; then; set intorde.=1; else; set intorde.=0;
    print j. adft. cr. intorde.;
  enddot;
endproc;
```

This set of commands belongs to the subprocedure "adftest" with "adfway" set to 2, 3.1 or 3.2. It tells TSP to calculate the appropriate critical values ("critv") and performs the (A)DF tests. The order of the (A)DF test "j", the t-value "adft", the critival value "cr" and the order of integration "intorde" are printed.

```
proc philtest dy,y,qadf0,qadf1,qadf2;
  olsq dy y(-1);
    set nob1=@nob; set s1=sqrt(@s2); set adft1=@t(1); genr res1=@res; set ses1=@ses(1);
  olsq dy y(-1) c;
    set nob2=@nob; set s2=sqrt(@s2); set adft2=@t(1); genr res2=@res; set ses2=@ses(1);
  olsq dy y(-1) c time;
    set nob3=@nob; set s3=sqrt(@s2); set adft3=@t(1); genr res3=@res; set ses3=@ses(1);
  test3;
endproc;
```

This subprocedure tells TSP to carry out PP tests. It estimates the OLS DF-test regressions, saves some key variables and refers to the subprocedure "test3" to calculate the z-statistics and carry out the actual tests.

```
proc test3;
 critv cr1 cr2 cr3;
 dot 1 2 3;
  cova(noprint) res.; set g0=@cova(1);
  set r=0;set w=0;
  if qway < 4; then; do;
    if qway = 1; then; set q=corrorde;
    if qway = 2; then; do;
      if nudiff=2; then; set q=qadf2;
      if nudiff=1; then; set q=qadf1;
      if nudiff=0; then; set q=qadf0; enddo;
    if qway = 3.0; then; set q=0;
    if qway = 3.1; then; set q=int(4*((nob./100)**0.25));
    if qway = 3.2; then; set q=int(12*((nob./100)**0.25));
    if q=0; then; set z.=adft.;
    if q>0 .and. q<10000; then; do;
      set l2=g0;
      do i=1 to q;
        set startg=start+i; smpl startg numobs;
        set r=-i; genr resi=res.(r);
        cova(noprint) res. resi;
        set ga(i)=@cova(2,1); set g(i)=2*(1-(i/(q+1)))*ga(i);
        set l2=l2+g(i); mmake hoop q,i,r,g0,ga(i),g(i),l2;
      enddo;
      smpl start numobs;
      set l=sqrt(l2);
      set z.=sqrt(g0/l2)*adft.-(1/2)*(l2-g0)*(nob.*ses./s.)*(1/l);
    enddo;
    if q=10000; then; do; set intorde.=10000; print q intorde.; enddo;
      else; do;
      if z.>cr.; then; set intorde.=1; else; set intorde.=0;
      print q z. cr. intorde.; enddo;
  enddo;
  if qway = 4; then; do;
    do q=1 to 12;
      set l2=g0;
      do i=1 to q;
        set startg=start+i; smpl startg numobs;
        set r=-i; genr resi=res.(r);
```

```
        cova(noprint) res. resi;
        set ga(i)=@cova(2,1); set g(i)=2*(1-(i/(q+1)))*ga(i);
        set l2=l2+g(i); mmake hoop q,i,r,g0,ga(i),g(i),l2;
      enddo;
      smpl start numobs;
      set l=sqrt(l2);
      set z.=sqrt(g0/l2)*adft.-(1/2)*(l2-g0)*(nob.*ses./s.)*(1/l);
      set ql2(q)=l2; set qz.(q)=z.; mmake hoop2 l l2 ql2(q) z. qz.(q);
    enddo;
    mmake z.ser adft.,qz.(1)-qz.(12); set w=0; set crq=cr.; copy z.ser zser;
    dot 1 2 3 4 5 6 7 8 9 10 11 12 13;
      set q.q.=w; set w=w+1; set cr.q.=crq;
      if zser(w)>crq; then; set int.q.=1; else; set int.q.=0;
    enddot;
    mmake vect q.q1-q.q13,adft.,qz.(1)-qz.(12),cr.q1-cr.q13,int.q1-int.q13;
    mform(type=general,nrow=13,ncol=4) results.=vect;
    print results.;
  enddo;
 enddot;
endproc;
```

This subprocedure is part of the subprocedure "philtest": it calculates the z-statistics and performs PP tests. First, it computes critical values. Second, it determines the number of autocovariances deemed relevant according to the user's choice of "qway". Third, it finds the autocovariances "ga". Fourth, it assigns each of these its Newey-West weight. Fifth, it calculates "l2", the value of λ^2. Sixth, it finds the value of the z-statistic ("z"). If "qway" is 4, this is repeated for q ranging from 0 to 12. Seventh, it carries out the actual PP tests and prints the outcome: the value of q, the z-statistic "z", the critical value "cr" and the order of integration "intorde". If "qway" is 4, for each of the three variants a matrix "results." is printed, with 13 rows corresponding to the 13 values q takes.

```
proc kpsstest dy,y,qadf0,qadf1,qadf2;
   olsq y c;
     genr res2=@res; set nob2=@nob;
   olsq y c time;
     genr res3=@res; set nob3=@nob;
   test4;
endproc;
```

This subprocedure asks TSP to perform KPSS tests. It estimates the appropriate OLS regressions, saves some variables needed further on and refers to the subprocedure "test4" for the calculation of the test statistics and for the actual tests.

```
proc test4;
 if sign=0.01; then; do;
   set crkpss2=0.74375-0.99187/nob2;
   set crkpss3=0.21778-0.235089/nob3; enddo;
 if sign=0.05; then; do;
   set crkpss2=0.46119+0.45911/nob2;
   set crkpss3=0.14795+0.035327/nob3; enddo;
 if sign=0.10; then; do;
   set crkpss2=0.34732+0.20695/nob2;
   set crkpss3=0.119298+0.100804/nob3; enddo;
```

```
dot 2 3;
 cova(noprint) res.; set g0=@cova(1);
 set r=0; set w=0; set sumst2=0; set st2=0;
 do i=startf to numobs;
   smpl startf i;
   msd(noprint) res.; set st2=@sum(1)**2; set sumst2=sumst2+st2;
 enddo;
 smpl startf numobs;
 if qway < 4; then; do;
   if qway = 1; then; set q=corrorde;
   if qway = 2; then; do;
     if nudiff=2; then; set q=qadf2;
     if nudiff=1; then; set q=qadf1;
     if nudiff=0; then; set q=qadf0; enddo;
   if qway = 3.0; then; set q=0;
   if qway = 3.1; then; set q=int(4*((nob./100)**0.25));
   if qway = 3.2; then; set q=int(12*((nob./100)**0.25));
   if q=0; then; set l2=g0;
   if q>0 .and. q<10000; then; do;
     set l2=g0;
     do i=1 to q;
       set startg=startf+i; smpl startg numobs;
       set r=-i; genr resi=res.(r); cova(noprint) res. resi;
       set ga(i)=@cova(2,1); set g(i)=2*(1-(i/(q+1)))*ga(i);
       set l2=l2+g(i); mmake hoop q,i,r,g0,ga(i),g(i),l2;
     enddo;
     smpl startf numobs;
   enddo;
   set n.=(1/(nob.**2))*sumst2*(1/l2);
   if q=10000; then; do; set intorde.=10000; print q intorde.; enddo;
     else; do;
       if n.>crkpss.; then; set intorde.=1; else; set intorde.=0;
       print q n. crkpss. intorde.; enddo;
 enddo;
 if qway = 4; then; do;
   set n.0=(1/(nob.**2))*(1/g0)*sumst2;
   do q=1 to 12;
     set l2=g0;
     do i=1 to q;
       set startg=startf+i; smpl startg numobs;
       set r=-i; genr resi=res.(r); cova(noprint) res. resi;
       set ga(i)=@cova(2,1); set g(i)=2*(1-(i/(q+1)))*ga(i);
       set l2=l2+g(i); mmake hoop q,i,r,g0,ga(i),g(i),l2;
     enddo;
     smpl startf numobs;
     set n.=(1/(nob.**2))*(1/l2)*sumst2;
     set ql2(q)=l2; set qn.(q)=n.; mmake hoop2 l2 ql2(q) n. qn.(q);
   enddo;
   mmake n.ser n.0,qn.(1)-qn.(12);
   set w=0; set crq=crkpss.; copy n.ser nser;
   dot 1 2 3 4 5 6 7 8 9 10 11 12 13;
     set q.q.=w; set w=w+1; set cr.q.=crq;
```

```
      if nser(w)>crq; then; set int.q.=1; else; set int.q.=0;
    enddot;
    mmake vect q.q1-q.q13,n.0,qn.(1)-qn.(12),cr.q1-cr.q13,int.q1-int.q13;
    mform(type=general,nrow=13,ncol=4) results.=vect;
    print results.;
  enddo;
 enddot;
endproc;
```

*These commands are part of the procedure "kpsstest". First, they find the appropriate Sephton (1995) critical values ("crkpss."). Second, the value of ΣS_t^2 is calculated. Third, the commands determine the number of autocovariances deemed relevant according to the user's choice of "qway" and find "l2", the value of λ^2 (see subprocedure "ztest"). Fourth, they calculate the value of the n-statistic ("n"). If "qway" is 4, this is repeated for q ranging from 0 to 12. Fifth, they perform the tests and report the results: q, the n-statistic "n", the critical value "crkpss" and the order of integration "intorde". If "qway" is 4, this is done for 13 values of q, viz. q=0 to q=12, so that for each variant of the KPSS tests a (13*4) matrix is printed (viz. "results2" for the null of level stationarity and "results3" for the null of trend stationarity).*

5. Output of the procedure for some macroeconomic variables

We use the TSP-procedure to investigate the order of integration of some macroeconomic time series. We select the following variables:
a. Canadian and U.S. narrow money supply, seasonally adjusted (CAM1S and USM1S)
b. Canadian and U.S. consumption price indices (CACPI and USCPI)
c. Canadian and U.S. long interest rates (CAIL and USIL)
d. the U.S. dollar exchange rate *vis-à-vis* the Canadian dollar (ECA).

The sample is 1973:2-1994:1. All data are quarterly and transformed into logs. They were taken from various issues of the IMF's *International Financial Statistics* (IFS-lines are 34b, 64, 60c, 61 and ae resp.).

We have included series with different a-priori trend behaviour. Firstly, we examine two variables that we suppose to be (positively) trending (a. and b.). Secondly, nothing in economic theory implies that interest rate series ought to be trending (c.). Finally, we have no priors concerning the trend behaviour of exchange rates (d.). Graphs of the (un-logged) series in levels, first and second differences are given in Appendix A.

We load the data (no frequency, sample 1-84) and the procedure into TSP and supply the commands

```
dot camls usmls cacpi uscpi cail usil eca;
 inttest l.,84,0.05,1,1,3.1,4,8;
 inttest l.,84,0.05,2,1,3.1,4,8;
 inttest l.,84,0.05,3,1,3.1,4,8;
enddot;
```

This means we ask TSP to perform (A)DF tests (*opt* = 1), PP tests (*opt* = 2) and KPSS tests (*opt* = 3), in this order, on each of the seven variables (*numobs* = 84) and to

- use the 5%-significance level (*sign* = 0.05);
- set *adfway* equal to 1, i.e. start from the ordinary DF test regression and add as many lagged differences of the series to be examined as necessary to remove serial correlation[8];
- set *qway* equal to 3.1, i.e. determine *q*, the number of autocovariances deemed relevant, according to the rule q = integer $[4.(T/100)^{0.25}]$; in this example, this boils down to setting *q* equal to 3;
- check for serial correlation up to the fourth order (*corrorde* = 4) and
- take 8 as the maximum ADF-order (*maxorder* = 8).

Appendix B contains the entire output file of the TSP-procedure in table format.

The conclusions w.r.t. the orders of integration are summarised in the table on the next page[9]. In general, we see that the three tests sometimes lead to different conclusions regarding the order of integration. Also, the results of the three (two) variants for each unit root (stationarity) test are not always in line with one another.

---

[8] Hassler & Wolters (1995, 39) provide evidence that ADF-tests can change "from significance to insignificance" when the number of included lags is increased. As ADF tests with "adfway" set to 1 typically result in a lower autoregressive order than when "adfway" is set to 2, we prefer the former to the latter.
In the case of e.g. the Canadian money supply the number of lags indeed in a number of cases differs substantially between both options, sometimes leading to conflicting conclusions w.r.t. the order of integration (The following results for the Canadian money supply in first differences are instructive: when "adfway" is 1, the number of lags is 0 or 1 and the order of integration is invariably 0, while when "adfway" is 2, 7 lags are included and the order of integration is 1 for all three (A)DF variants).
The output of the procedure when "adfway" is set to 2, is available from the author on request.

[9]. All other results we refer to in this section, are available from the author.

a. According to almost all test versions, both money supplies (second and third row of the table) can be approximated by an I(1) process. In the case of Canadian money, only the KPSS test points to the presence of a double unit root if allowance for a trend is made (variant (3)). As noted in section 3, these contradictory results might be due to a type I error[10].

For the U.S. money stock, two test results refute the I(1) conclusion. On the one hand, the KPSS-with-trend outcome is that this variable is trend stationary instead of integrated of order 1. As mentioned in section 3, low power of the tests used is what we should conclude from this[11]. On the other hand, the U.S. money supply is I(2) according to the "no constant, no trend" variant of the ADF tests. As this variant is hardly appropriate for dealing with macroeconomic time series, we will give more weight to the other test results and conclude that, on balance, an I(1) process for both money stocks provides a reasonable approximation[12].

b. There is some doubt in the empirical literature as to whether the consumption price indices are I(2) rather than I(1). This means that the inflation rates (approximated by the first difference of the logged CPI) are integrated of order one or (trend) stationary. The results in the table suggest that, according to both PP and KPSS tests the inflation rates are trend stationary, whereas the ADF tests detect I(1) behaviour in the case of the U.S. inflation rate and allow for both approximations in the case of

---

10 Decreasing the probability of a type-I error by lowering the significance level to 1% indeed leads to the conclusion of a single unit root in the Canadian money supply when a trend term is included in the KPSS test regression and to that of a stationary money stock when it is left out (obviously less appropriate when dealing with a trending variable). Moreover, a sensitivity analysis of the KPSS outcomes to different q-values with a 0.05 significance level makes clear that the KPSS cum trend test does not reject the null of stationarity of the Canadian money stock in first differences when q is 4 or higher (note: q is 3 in the table).

11 The power of the tests can be increased by choosing a higher significance level. When $\alpha = 0.10$, both KPSS tests indeed point to a single unit root in the USM1S variable.

12 This result for the U.S. money stock is in line with MacDonald and Taylor (1991), who use monthly data of the same source over the period 1976:1 - 1990:12.

Table: Results of unit root and stationarity tests

	ADF adfway = 1			PP q = 3			KPSS q = 3	
	(1)	(2)	(3)	(1)	(2)	(3)	(2)	(3)
CAM1S								
Nudiff=2	0(7)	0(3,7)	0(3,7)	0	0	0	0	0
Nudiff=1	0(1)	0(0,1)	0(0,1)	0	0	0	0	1
Nudiff=0	1(0)	1(0)	1(0)	1	1	1	1	1
USM1S								
Nudiff=2	0(1,2)	0(2)	0(2)	0	0	0	0	0
Nudiff=1	1(1)	0(1)	0(1)	0	0	0	0	0
Nudiff=0	1(2)	1(2)	1(2)	1	1	1	1	0
CACPI								
Nudiff=2	0(1)	0(1)	0(1)	0	0	0	0	0
Nudiff=1	1(2)	1(2)	0(0,1) 1(2)	1	1	0	1	0
Nudiff=0	1(3)	0(1,2) 1(3)	1(1,2,3)	1	0	1	1	1
USCPI								
Nudiff=2	0(1)	0(1)	0(1)	0	0	0	0	0
Nudiff=1	1(2)	1(2)	1(2)	1	1	0	1	0
Nudiff=0	1(3)	1(3)	1(3)	1	0	1	1	1
CAIL								
Nudiff=2	0(2)	0(2)	0(2)	0	0	0	0	0
Nudiff=1	0(0)	0(0)	0(0)	0	0	0	0	0
Nudiff=0	1(0)	1(0)	1(0)	1	1	1	0	1
USIL								
Nudiff=2	0(1)	0(1)	0(1)	0	0	0	0	0
Nudiff=1	0(0)	0(0)	0(0)	0	0	0	0	0
Nudiff=0	1(1)	1(1)	1(0,1)	1	1	1	0	1
ECA								
Nudiff=2	0(4)	0(4)	0(4)	0	0	0	0	0
Nudiff=1	0(0)	0(0)	0(0)	0	0	0	0	0
Nudiff=0	1(0)	1(0)	1(0)	1	1	1	1	1

Notes: The results in this table were taken from Appendix B. The significance level is 5%; all data are in logs (data definitions in main text). Nudiff points to the series under investigation: it is 2 for the series of double first differences, 1 for first differences and 0 for the variable in (log) levels.
In the head of the table (1), (2) and (3) refer to the ADF or PP test variant "no constant, no trend", "constant, no trend" and "constant and trend" resp. In the last column (2) and (3) refer to the KPSS tests of the null hypotheses of level and trend stationarity resp. See main text for the explanation of "adfway = 1" and "q = 3" in the head of the table.
In the body of the table 0 and 1 refer to the order of integration. In the second column the figure(s) in brackets give(s) the (A)DF order(s) (0 refers to the non-augmented DF test).

the Canadian variable[13].

We investigate the sensitivity of these results to changes in two parameters, viz. q, the number of autocovariances included in the estimate of the error variance (needed in the PP and KPSS test statistics), and α, the significance level. Increasing q does not change the conclusions from the PP and KPSS tests. Using a lower significance level leads to PP tests in all versions in favour of a US inflation rate with a double unit root. With a higher significance level, however, only the "no constant, no trend" version of the PP tests supports an I(2) US inflation rate. The other results are unchanged. We would therefore tend to regard both inflation rates as variables that can be approximated by a unit root as well as a trend stationary process, though the former might be more appropriate for the U.S. variable.

Note also the importance of the Dickey & Pantula (1987) suggestion to test for a descending order of integration: according to e.g. the PP tests with constant only, both CPIs are I(0), while the inflation rates have a unit root. Clearly this would be inconsistent.

c. Regarding the interest rates in table 1 there is rather clear-cut evidence that they are both integrated of order one[14]. The only conflicting outcome is the KPSS test result, which suggests that both interest rates in log levels are level stationary. Once again, this indicates the lack of power of unit root and stationarity tests (see section 3)[15].

d. Finally, the exchange rate Canadian dollar - U.S. dollar appears to contain one unit

---

13 Pippenger (1993) uses monthly data on producer price indices from the same source for the period 1973:1 - 1988:6 and finds that the U.S. inflation rate is integrated of order one. The null hypothesis of a unit root in the Canadian inflation rate can be rejected at the 10 percent level only. According to the results in MacDonald (1993) the U.S. CPI contains a double unit root, whereas the U.S. wholesale price index in first differences could be stationary around a trend; unit root behaviour of the Canadian inflation rate is narrowly rejected for both indices (same source, data period 1974:1 to 1990:6).

14 A comparable result is obtained by MacDonald & Taylor (1991, same data source, span 1976:1 - 1990:12), who find that the U.S. long rate is I(1) except in the case of PP tests including a time trend, where it is I(0).

15 Indeed, when we increase the power of the tests by switching to a higher significance level (α = 0.10), all KPSS tests also reveal I(1) behaviour of interest rates.

root. For once, all tests in all versions support this conclusion[16].

We believe that this discussion highlights that unit root and stationarity test outcomes are to be interpreted with care. Indeed, as noted in section 3, these tests ought to be used only to assess whether a random walk process is a justifiable approximation, e.g. before performing cointegration tests.

6. Conclusions

This paper proposes a TSP-procedure to perform unit root and stationarity tests with a simple one-line command as a richer alternative to the unit root testing strategies available in standard statistical packages. On the one hand, the TSP-procedure incorporates some recent developments not readily available in standard programs (e.g. PP and KPSS tests and a new set of critical KPSS values) as well as the often ignored Dickey & Pantula recommendation to test for a descending number of unit roots. On the other hand, the user can choose from an option menu to determine how the ADF-lag length is set and how the number of autocovariances in the Newey-West estimator is fixed (PP and KPSS tests).

Whereas we hope that the TSP-procedure makes unit root and stationarity tests easier to perform, the application in section 5 shows that the test results ought to be interpreted with caution. As noted in section 3, one should indeed bear in mind that, in the words of Hamilton (1994, 516), "the goal of unit root tests is to find a parsimonious representation that gives a reasonable approximation to the true process, as opposed to determining whether or not the true process is literally I(1)".

---

16 MacDonald (1993, supra) and Taylor (1988, 1973:6-1985:12) confirm this finding.

REFERENCES

CAMPBELL, John Y. and Pierre PERRON (1992), "Pitfalls and Opportunities : What Macroeconomists Should Know about Unit Roots", in Olivier Jean BLANCHARD and Stanley FISCHER, eds., NBER Macroeconomics Annual 1991, Cambridge, Mass., The MIT Press, 1992, p. 141-201.

CROMWELL, Jeff B., Walter C. LABYS and Michel TERRAZA (1994), Univariate Tests for Time Series Models, Thousand Oaks, Sage Publications, Quantitative Applications in the Social Sciences No. 99 (Sage University Papers), 96 p.

CUTHBERTSON, Keith, Stephen G. HALL and Mark P. TAYLOR (1992), Applied Econometric Techniques, New York, Philip Allan, 274 p.

DICKEY, David A. and Sastry G. PANTULA (1987), "Determining the Order of Differencing in Autoregressive Processes", Journal of Business and Economic Statistics, Vol. 5, No. 4, p. 455-461.

ENGLE, Robert F. and C.W.J. GRANGER (1987), "Cointegration and Error Correction : Representation, Estimation, and Testing", Econometrica, Vol. 55, No. 2, March, p. 251-276.

ENGLE, R.F. and C.W.J. GRANGER, eds. (1991), Long-Run Economic Relationships. Readings in Cointegration, Oxford, Oxford University Press, 301 p.

FULLER, Wayne A. (1976), Introduction to Statistical Time Series, New York, John Wiley & Sons, 470 p.

HAMILTON, James D. (1994), Time Series Analysis, Princeton, New Jersey, Princeton University Press, 799 p.

HARRIS, David and Brett INDER (1994), "A test of the null hypothesis of cointegration", Chapter 5 in Colin P. HARGREAVES (ed.), Nonstationary Time Series Analysis and Cointegration, Oxford, Oxford University Press.

HARRIS, Richard I.D. (1995), *Using Cointegration Analysis in Econometric Modelling*, Hemel Hempstead, Prentice Hall/Harvester Wheatsheaf, 176 p.

HARVEY, Andrew C. (1989), Forecasting, structural time series models and the Kalman filter, Cambridge, Cambridge University Press, 554 p.

HARVEY, A.C. (1990), The Econometric Analysis of Time Series, New York, Philip Allan, 2de ed., 387 p.

HASSLER, Uwe and Jürgen WOLTERS (1995), "Long Memory in Inflation Rates: International Evidence", Journal of Business and Economic Statistics, Vol. 13, No. 1, January, p. 37-45.

JUDGE, George D., R. Carter HILL, William E. GRIFFITHS, Helmut LÜTKEPOHL and Tsoung-Chao LEE (1988), Introduction to the Theory and Practice of Econometrics, New York, John Wiley & Sons, 2nd ed., 1024 p.

KWIATKOWSKI, Denis, Peter C.B. PHILLIPS, Peter SCHMIDT and Yongcheol SHIN (1992), "Testing the null hypothesis of stationarity against the alternative of a unit root. How sure are we that economic time series have a unit root?", Journal of Econometrics, Vol., p. 159-178.

MacDONALD, Ronald (1993), "Long-Run Purchasing Power Parity: Is It for Real?", Review of Economics and Statistics, Vol. 75, No. 4, p. 690-5.

MacDONALD, Ronald and Mark P. TAYLOR (1991), "The monetary approach to the exchange rate: Long-run relationships and coefficient restrictions", Economics Letters, Vol. 37, p. 179-85.

MacKINNON, James (1991), "Critical Values for Cointegration Tests", in R.F. ENGLE and C.W.J. GRANGER, eds. (1991), Long-Run Economic Relationships. Readings in Cointegration, Oxford, Oxford University Press, p. 267-276.

PERRON, Pierre (1988), "Trends and Random Walks in Macroeconomic Time Series. Further Evidence from a New Approach", Journal of Economic Dynamics and Control, Vol. 12, p. 297-332.

PHILLIPS, P.C.B. (1987), "Time Series Regression With a Unit Root", Econometrica, vol. 55, no. 2, p. 277-301.

PIPPENGER, Michael K. (1993), "Cointegration tests of purchasing power parity: the case of Swiss exchange rates", Journal of International Money and Finance, Vol. 12, p. 46-61.

SCHMIDT, Peter and Peter C.B. PHILLIPS (1992), "LM Test for a Unit Root in the Presence of Determininstic Trends", Oxford Bulletin of Economics and Statistics, Vol. 54, No. 3, p. 257-287.

SCHWERT, G.W. (1989), "Tests for unit roots: a Monte Carlo investigation", *Journal of Business and Economics Statistics*, Vol. 7, p. 147-59.

SEPHTON, Peter S. (1995), "Response surface estimates of the KPSS stationarity test", Economics Letters, Vol. 47, March, p. 255-261.

STEWART, Jon (1991), Econometrics, New York, Philip Allen, 331 p.

TAYLOR, Mark P. (1988), "An empirical examination of long-run purchasing power parity using cointegration techniques", Applied Economics, Vol. 20, p. 1369-81.

Appendix A: graphs of data

CAM1S 1973:2 1994:1

billions of Canadian dollars; end of quarter



DCAM1S 1973:3 1994:1

CAM1S in first differences



DDCAM1S 1973:4 1994:1

CAM1S in second differences



USM1S 1973:2 1994:1

billions of U.S. dollars; end of quarter



DUSM1S 1973:3 1994:1

USM1S in first differences



DDUSM1S 1973:4 1994:1

USM1S in second differences



CACPI 1973:2 1994:1



DCACPI 1973:3 1994:1



DDCACPI 1973:4 1994:1



USCPI 1973:2 1994:1



DUSCPI 1973:3 1994:1



DDUSCPI 1973:4 1994:1



CAIL 1973:2 1994:1



DCAIL 1973:3 1994:1



DDCAIL 1973:4 1994:1



USIL 1973:2 1994:1

long-term Government bond yield; percent per annum



DUSIL 1973:3 1994:1

USIL in first differences



DDUSIL 1973:4 1994:1

USIL in second differences



ECA 1973:2 1994:1

market rate Canadian $ per U.S. $; end of quarter



DECA 1973:3 1994:1

ECA in first differences



DDECA 1973:4 1994:1

ECA in second differences



Appendix B: Output of TSP-procedure

= Direct output of the TSP-procedure, except for:
a. the current sample (1 to 84), the value of "opt" (here: 1, 2 and 3), the significance level (SIGN=0.05), the equation numbers of the test regressions and the method of estimation (Ordinary Least Squares), all of which are suppressed.
b. the layout, which is modified into table form

CAM1S

NUDIFF	(A)DF tests	PP tests	KPSS tests
2	J = 3.00000 ADFT2 = -8.56114 CR2 = -2.89858 INTORDE2 = 0.00000 J = 3.00000 ADFT3 = -8.50627 CR3 = -3.46731 INTORDE3 = 0.00000 J = 7.00000 ADFT1 = -4.17862 CR1 = -1.94468 INTORDE1 = 0.00000 J = 7.00000 ADFT2 = -4.14710 CR2 = -2.90063 INTORDE2 = 0.00000 J = 7.00000 ADFT3 = -4.10170 CR3 = -3.47044 INTORDE3 = 0.00000	Q = 3.00000 Z1 = -24.59662 CR1 = -1.94421 INTORDE1 = 0.00000 Q = 3.00000 Z2 = -24.41946 CR2 = -2.89718 INTORDE2 = 0.00000 Q = 3.00000 Z3 = -24.23991 CR3 = -3.46518 INTORDE3 = 0.00000	Q = 3.00000 N2 = 0.036398 CRKPSS2 = 0.46679 INTORDE2 = 0.00000 Q = 3.00000 N3 = 0.032771 CRKPSS3 = 0.14838 INTORDE3 = 0.00000
1	J = 0.00000 ADFT2 = -9.00371 CR2 = -2.89673 INTORDE2 = 0.00000 J = 0.00000 ADFT3 = -8.94531 CR3 = -3.46451 INTORDE3 = 0.00000 28 29 30 J = 1.00000 ADFT1 = -3.44118 CR1 = -1.94421 INTORDE1 = 0.00000 J = 1.00000 ADFT2 = -5.35453 CR2 = -2.89718 INTORDE2 = 0.00000 J = 1.00000 ADFT3 = -5.32097 CR3 = -3.46518 INTORDE3 = 0.00000	Q = 3.00000 Z1 = -6.20818 CR1 = -1.94415 INTORDE1 = 0.00000 Q = 3.00000 Z2 = -9.05993 CR2 = -2.89673 INTORDE2 = 0.00000 Q = 3.00000 Z3 = -9.00701 CR3 = -3.46451 INTORDE3 = 0.00000	Q = 3.00000 N2 = 0.15437 CRKPSS2 = 0.46672 INTORDE2 = 0.00000 Q = 3.00000 N3 = 0.15415 CRKPSS3 = 0.14838 INTORDE3 = 1.00000
0	J = 0.00000 ADFT1 = 6.22458 CR1 = -1.94410 INTORDE1 = 1.00000 J = 0.00000 ADFT2 = -0.21140 CR2 = -2.89630 INTORDE2 = 1.00000 J = 0.00000 ADFT3 = -1.31576 CR3 = -3.46385 INTORDE3 = 1.00000	Q = 3.00000 Z1 = 5.63410 CR1 = -1.94410 INTORDE1 = 1.00000 Q = 3.00000 Z2 = -0.23549 CR2 = -2.89630 INTORDE2 = 1.00000 Q = 3.00000 Z3 = -1.46563 CR3 = -3.46385 INTORDE3 = 1.00000	Q = 3.00000 N2 = 2.11313 CRKPSS2 = 0.46666 INTORDE2 = 1.00000 Q = 3.00000 N3 = 0.21316 CRKPSS3 = 0.14837 INTORDE3 = 1.00000

USM1S

NUDIFF	(A)DF tests	PP tests	KPSS tests
2	J = 1.00000 ADFT1 = -10.27613 CR1 = -1.94428 INTORDE1 = 0.00000 J = 2.00000 ADFT1 = -8.25246 CR1 = -1.94434 INTORDE1 = 0.00000 J = 2.00000 ADFT2 = -8.25284 CR2 = -2.89810 INTORDE2 = 0.00000 J = 2.00000 ADFT3 = -8.20886 CR3 = -3.46658 INTORDE3 = 0.00000	Q = 3.00000 Z1 = -22.00087 CR1 = -1.94421 INTORDE1 = 0.00000 Q = 3.00000 Z2 = -21.85877 CR2 = -2.89718 INTORDE2 = 0.00000 Q = 3.00000 Z3 = -21.76676 CR3 = -3.46518 INTORDE3 = 0.00000	Q = 3.00000 N2 = 0.032955 CRKPSS2 = 0.46679 INTORDE2 = 0.00000 Q = 3.00000 N3 = 0.032372 CRKPSS3 = 0.14838 INTORDE3 = 0.00000
1	J = 1.00000 ADFT1 = -1.37829 CR1 = -1.94421 INTORDE1 = 1.00000 J = 1.00000 ADFT2 = -3.74690 CR2 = -2.89718 INTORDE2 = 0.00000 J = 1.00000 ADFT3 = -3.99032 CR3 = -3.46518 INTORDE3 = 0.00000	Q = 3.00000 Z1 = -2.66253 CR1 = -1.94415 INTORDE1 = 0.00000 Q = 3.00000 Z2 = -7.57540 CR2 = -2.89673 INTORDE2 = 0.00000 Q = 3.00000 Z3 = -7.85252 CR3 = -3.46451 INTORDE3 = 0.00000	Q = 3.00000 N2 = 0.26681 CRKPSS2 = 0.46672 INTORDE2 = 0.00000 Q = 3.00000 N3 = 0.10406 CRKPSS3 = 0.14838 INTORDE3 = 0.00000
0	J = 2.00000 ADFT1 = 3.62480 CR1 = -1.94421 INTORDE1 = 1.00000 J = 2.00000 ADFT2 = 1.18227 CR2 = -2.89718 INTORDE2 = 1.00000 J = 2.00000 ADFT3 = -2.30436 CR3 = -3.46518 INTORDE3 = 1.00000	Q = 3.00000 Z1 = 9.98603 CR1 = -1.94410 INTORDE1 = 1.00000 Q = 3.00000 Z2 = 1.48469 CR2 = -2.89630 INTORDE2 = 1.00000 Q = 3.00000 Z3 = -1.93527 CR3 = -3.46385 INTORDE3 = 1.00000	Q = 3.00000 N2 = 2.13887 CRKPSS2 = 0.46666 INTORDE2 = 1.00000 Q = 3.00000 N3 = 0.12838 CRKPSS3 = 0.14837 INTORDE3 = 0.00000

CACPI

NUDIFF	(A)DF tests	PP tests	KPSS tests
2	J = 1.00000 ADFT1 = -9.46203 CR1 = -1.94428 INTORDE1 = 0.00000 J = 1.00000 ADFT2 = -9.48304 CR2 = -2.89763 INTORDE2 = 0.00000 J = 1.00000 ADFT3 = -9.45428 CR3 = -3.46587 INTORDE3 = 0.00000	Q = 3.00000 Z1 = -12.85520 CR1 = -1.94421 INTORDE1 = 0.00000 Q = 3.00000 Z2 = -12.79764 CR2 = -2.89718 INTORDE2 = 0.00000 Q = 3.00000 Z3 = -12.76817 CR3 = -3.46518 INTORDE3 = 0.00000	Q = 3.00000 N2 = 0.060562 CRKPSS2 = 0.46679 INTORDE2 = 0.00000 Q = 3.00000 N3 = 0.043887 CRKPSS3 = 0.14838 INTORDE3 = 0.00000
1	J = 0.00000 ADFT3 = -4.63616 CR3 = -3.46451 INTORDE3 = 0.00000 J = 1.00000 ADFT3 = -3.91126 CR3 = -3.46518 INTORDE3 = 0.00000 J = 2.00000 ADFT1 = -1.33159 CR1 = -1.94428 INTORDE1 = 1.00000 J = 2.00000 ADFT2 = -1.26091 CR2 = -2.89763 INTORDE2 = 1.00000 J = 2.00000 ADFT3 = -2.89010 CR3 = -3.46587 INTORDE3 = 1.00000	Q = 3.00000 Z1 = -1.60047 CR1 = -1.94415 INTORDE1 = 1.00000 Q = 3.00000 Z2 = -2.27042 CR2 = -2.89673 INTORDE2 = 1.00000 Q = 3.00000 Z3 = -4.54882 CR3 = -3.46451 INTORDE3 = 0.00000	Q = 3.00000 N2 = 1.48025 CRKPSS2 = 0.46672 INTORDE2 = 1.00000 Q = 3.00000 N3 = 0.10030 CRKPSS3 = 0.14838 INTORDE3 = 0.00000
0	J = 1.00000 ADFT2 = -3.35864 CR2 = -2.89673 INTORDE2 = 0.00000 J = 1.00000 ADFT3 = 0.28153 CR3 = -3.46451 INTORDE3 = 1.00000 J = 2.00000 ADFT2 = -3.32589 CR2 = -2.89718 INTORDE2 = 0.00000 J = 2.00000 ADFT3 = -0.14692 CR3 = -3.46518 INTORDE3 = 1.00000 J = 3.00000 ADFT1 = 0.47021 CR1 = -1.94428 INTORDE1 = 1.00000 J = 3.00000 ADFT2 = -2.60163 CR2 = -2.89763 INTORDE2 = 1.00000 J = 3.00000 ADFT3 = -0.25895 CR3 = -3.46587 INTORDE3 = 1.00000	Q = 3.00000 Z1 = 7.65226 CR1 = -1.94410 INTORDE1 = 1.00000 Q = 3.00000 Z2 = -5.94007 CR2 = -2.89630 INTORDE2 = 0.00000 Q = 3.00000 Z3 = 0.26153 CR3 = -3.46385 INTORDE3 = 1.00000	Q = 3.00000 N2 = 2.07700 CRKPSS2 = 0.46666 INTORDE2 = 1.00000 Q = 3.00000 N3 = 0.51993 CRKPSS3 = 0.14837 INTORDE3 = 1.00000

USCPI

NUDIFF	(A)DF tests	PP tests	KPSS tests
2	J = 1.00000 ADFT1 = -11.63667 CR1 = -1.94428 INTORDE1 = 0.00000 J = 1.00000 ADFT2 = -11.63159 CR2 = -2.89763 INTORDE2 = 0.00000 J = 1.00000 ADFT3 = -11.55572 CR3 = -3.46587 INTORDE3 = 0.00000	Q = 3.00000 Z1 = -12.01115 CR1 = -1.94421 INTORDE1 = 0.00000 Q = 3.00000 Z2 = -11.93371 CR2 = -2.89718 INTORDE2 = 0.00000 Q = 3.00000 Z3 = -11.83880 CR3 = -3.46518 INTORDE3 = 0.00000	Q = 3.00000 N2 = 0.043328 CRKPSS2 = 0.46679 INTORDE2 = 0.00000 Q = 3.00000 N3 = 0.043361 CRKPSS3 = 0.14838 INTORDE3 = 0.00000
1	J = 2.00000 ADFT1 = -1.47914 CR1 = -1.94428 INTORDE1 = 1.00000 J = 2.00000 ADFT2 = -1.68548 CR2 = -2.89763 INTORDE2 = 1.00000 J = 2.00000 ADFT3 = -2.13722 CR3 = -3.46587 INTORDE3 = 1.00000	Q = 3.00000 Z1 = -1.44869 CR1 = -1.94415 INTORDE1 = 1.00000 Q = 3.00000 Z2 = -2.74664 CR2 = -2.89673 INTORDE2 = 1.00000 Q = 3.00000 Z3 = -3.91100 CR3 = -3.46451 INTORDE3 = 0.00000	Q = 3.00000 N2 = 1.08828 CRKPSS2 = 0.46672 INTORDE2 = 1.00000 Q = 3.00000 N3 = 0.11046 CRKPSS3 = 0.14838 INTORDE3 = 0.00000
0	J = 3.00000 ADFT1 = 0.92587 CR1 = -1.94428 INTORDE1 = 1.00000 J = 3.00000 ADFT2 = -1.68465 CR2 = -2.89763 INTORDE2 = 1.00000 J = 3.00000 ADFT3 = -1.51824 CR3 = -3.46587 INTORDE3 = 1.00000	Q = 3.00000 Z1 = 7.48500 CR1 = -1.94410 INTORDE1 = 1.00000 Q = 3.00000 Z2 = -4.38730 CR2 = -2.89630 INTORDE2 = 0.00000 Q = 3.00000 Z3 = -1.00110 CR3 = -3.46385 INTORDE3 = 1.00000	Q = 3.00000 N2 = 2.06013 CRKPSS2 = 0.46666 INTORDE2 = 1.00000 Q = 3.00000 N3 = 0.49606 CRKPSS3 = 0.14837 INTORDE3 = 1.00000

CAIL

NUDIFF	(A)DF tests	PP tests	KPSS tests
2	J = 2.00000 ADFT1 = -9.54519 CR1 = -1.94434 INTORDE1 = 0.00000 J = 2.00000 ADFT2 = -9.49681 CR2 = -2.89810 INTORDE2 = 0.00000 J = 2.00000 ADFT3 = -9.42668 CR3 = -3.46658 INTORDE3 = 0.00000	Q = 3.00000 Z1 = -16.91613 CR1 = -1.94421 INTORDE1 = 0.00000 Q = 3.00000 Z2 = -16.78146 CR2 = -2.89718 INTORDE2 = 0.00000 Q = 3.00000 Z3 = -16.64705 CR3 = -3.46518 INTORDE3 = 0.00000	Q = 3.00000 N2 = 0.032644 CRKPSS2 = 0.46679 INTORDE2 = 0.00000 Q = 3.00000 N3 = 0.029901 CRKPSS3 = 0.14838 INTORDE3 = 0.00000
1	J = 0.00000 ADFT1 = -7.71195 CR1 = -1.94415 INTORDE1 = 0.00000 J = 0.00000 ADFT2 = -7.66456 CR2 = -2.89673 INTORDE2 = 0.00000 J = 0.00000 ADFT3 = -7.92167 CR3 = -3.46451 INTORDE3 = 0.00000	Q = 3.00000 Z1 = -7.67046 CR1 = -1.94415 INTORDE1 = 0.00000 Q = 3.00000 Z2 = -7.62137 CR2 = -2.89673 INTORDE2 = 0.00000 Q = 3.00000 Z3 = -7.86510 CR3 = -3.46451 INTORDE3 = 0.00000	Q = 3.00000 N2 = 0.35728 CRKPSS2 = 0.46672 INTORDE2 = 0.00000 Q = 3.00000 N3 = 0.047045 CRKPSS3 = 0.14838 INTORDE3 = 0.00000
0	J = 0.00000 ADFT1 = -0.14442 CR1 = -1.94410 INTORDE1 = 1.00000 J = 0.00000 ADFT2 = -1.43040 CR2 = -2.89630 INTORDE2 = 1.00000 J = 0.00000 ADFT3 = -1.39220 CR3 = -3.46385 INTORDE3 = 1.00000	Q = 3.00000 Z1 = -0.15240 CR1 = -1.94410 INTORDE1 = 1.00000 Q = 3.00000 Z2 = -1.61210 CR2 = -2.89630 INTORDE2 = 1.00000 Q = 3.00000 Z3 = -1.47505 CR3 = -3.46385 INTORDE3 = 1.00000	Q = 3.00000 N2 = 0.38997 CRKPSS2 = 0.46666 INTORDE2 = 0.00000 Q = 3.00000 N3 = 0.38981 CRKPSS3 = 0.14837 INTORDE3 = 1.00000

USIL

NUDIFF	(A)DF tests	PP tests	KPSS tests
2	J = 1.00000 ADFT1 = -10.08933 CR1 = -1.94428 INTORDE1 = 0.00000 J = 1.00000 ADFT2 = -10.02355 CR2 = -2.89763 INTORDE2 = 0.00000 J = 1.00000 ADFT3 = -9.96034 CR3 = -3.46587 INTORDE3 = 0.00000	Q = 3.00000 Z1 = -15.73585 CR1 = -1.94421 INTORDE1 = 0.00000 Q = 3.00000 Z2 = -15.62105 CR2 = -2.89718 INTORDE2 = 0.00000 Q = 3.00000 Z3 = -15.51197 CR3 = -3.46518 INTORDE3 = 0.00000	Q = 3.00000 N2 = 0.056362 CRKPSS2 = 0.46679 INTORDE2 = 0.00000 Q = 3.00000 N3 = 0.032007 CRKPSS3 = 0.14838 INTORDE3 = 0.00000
1	J = 0.00000 ADFT1 = -7.00242 CR1 = -1.94415 INTORDE1 = 0.00000 J = 0.00000 ADFT2 = -6.96023 CR2 = -2.89673 INTORDE2 = 0.00000 J = 0.00000 ADFT3 = -7.17921 CR3 = -3.46451 INTORDE3 = 0.00000	Q = 3.00000 Z1 = -7.03511 CR1 = -1.94415 INTORDE1 = 0.00000 Q = 3.00000 Z2 = -6.99409 CR2 = -2.89673 INTORDE2 = 0.00000 Q = 3.00000 Z3 = -7.18319 CR3 = -3.46451 INTORDE3 = 0.00000	Q = 3.00000 N2 = 0.38789 CRKPSS2 = 0.46672 INTORDE2 = 0.00000 Q = 3.00000 N3 = 0.046997 CRKPSS3 = 0.14838 INTORDE3 = 0.00000
0	J = 0.00000 ADFT3 = -1.25998 CR3 = -3.46385 INTORDE3 = 1.00000 J = 1.00000 ADFT1 = -0.25004 CR1 = -1.94415 INTORDE1 = 1.00000 J = 1.00000 ADFT2 = -1.38278 CR2 = -2.89673 INTORDE2 = 1.00000 J = 1.00000 ADFT3 = -1.46475 CR3 = -3.46451 INTORDE3 = 1.00000	Q = 3.00000 Z1 = -0.18522 CR1 = -1.94410 INTORDE1 = 1.00000 Q = 3.00000 Z2 = -1.40624 CR2 = -2.89630 INTORDE2 = 1.00000 Q = 3.00000 Z3 = -1.43376 CR3 = -3.46385 INTORDE3 = 1.00000	Q = 3.00000 N2 = 0.45518 CRKPSS2 = 0.46666 INTORDE2 = 0.00000 Q = 3.00000 N3 = 0.43959 CRKPSS3 = 0.14837 INTORDE3 = 1.00000

ECA

NUDIFF	(A)DF tests	PP tests	KPSS tests
2	J = 4.00000 ADFT1 = -6.87417 CR1 = -1.94447 INTORDE1 = 0.00000 J = 4.00000 ADFT2 = -6.84151 CR2 = -2.89907 INTORDE2 = 0.00000 J = 4.00000 ADFT3 = -6.80597 CR3 = -3.46806 INTORDE3 = 0.00000	Q = 3.00000 Z1 = -18.97114 CR1 = -1.94421 INTORDE1 = 0.00000 Q = 3.00000 Z2 = -18.83008 CR2 = -2.89718 INTORDE2 = 0.00000 Q = 3.00000 Z3 = -18.69040 CR3 = -3.46518 INTORDE3 = 0.00000	Q = 3.00000 N2 = 0.057184 CRKPSS2 = 0.46679 INTORDE2 = 0.00000 Q = 3.00000 N3 = 0.034481 CRKPSS3 = 0.14838 INTORDE3 = 0.00000
1	J = 0.00000 ADFT1 = -7.86604 CR1 = -1.94415 INTORDE1 = 0.00000 J = 0.00000 ADFT2 = -8.05583 CR2 = -2.89673 INTORDE2 = 0.00000 J = 0.00000 ADFT3 = -7.99621 CR3 = -3.46451 INTORDE3 = 0.00000	Q = 3.00000 Z1 = -7.90836 CR1 = -1.94415 INTORDE1 = 0.00000 Q = 3.00000 Z2 = -8.12989 CR2 = -2.89673 INTORDE2 = 0.00000 Q = 3.00000 Z3 = -8.07364 CR3 = -3.46451 INTORDE3 = 0.00000	Q = 3.00000 N2 = 0.12896 CRKPSS2 = 0.46672 INTORDE2 = 0.00000 Q = 3.00000 N3 = 0.12357 CRKPSS3 = 0.14838 INTORDE3 = 0.00000
0	J = 0.00000 ADFT1 = 0.84666 CR1 = -1.94410 INTORDE1 = 1.00000 J = 0.00000 ADFT2 = -1.05022 CR2 = -2.89630 INTORDE2 = 1.00000 J = 0.00000 ADFT3 = -1.33511 CR3 = -3.46385 INTORDE3 = 1.00000	Q = 3.00000 Z1 = 0.63139 CR1 = -1.94410 INTORDE1 = 1.00000 Q = 3.00000 Z2 = -1.16155 CR2 = -2.89630 INTORDE2 = 1.00000 Q = 3.00000 Z3 = -1.51230 CR3 = -3.46385 INTORDE3 = 1.00000	Q = 3.00000 N2 = 1.20379 CRKPSS2 = 0.46666 INTORDE2 = 1.00000 Q = 3.00000 N3 = 0.38400 CRKPSS3 = 0.14837 INTORDE3 = 1.00000

LIJST VAN RECENTE SESO-RAPPORTEN

VANDENBUSSCHE H., *How can Japanese and Central-European exporters to the EU avoid antidumping duties ?,* December 1994, 33 p. **(94/314)**

DE BORGER B., FERRIER G. en K. KERSTENS, *The choice of a technical efficiency measure on the free disposal hull reference technology : a comparison using US banking data,* January 1995, 34 p. **(95/315)**

KONINGS J. en H. VANDENBUSSCHE, *How does increased foreign competition affect UK employment and wages ? An empirical exploration,* February 1995, 19 p. **(95/316)**

HEYLEN F. en A. VAN POECK, *Central bank independence : only part of the inflation story,* February 1995, 20 p. **(95/317)**

VANHOUDT P., *Het belang van investeringen in mensen, materieel en know-how voor economische groei : empirische evidentie,* april 1995, 29 p. **(95/318)**

NONNEMAN W. en P. VANHOUDT, *A further augmentation of the Solow model and the empirics of economic growth for OECD countries,* April 1995, 12 p. **(95/319)**

MEERSMAN H. en D. CASSIMON, *Investment in the EC-countries : does reducing uncertainty matter ?,* April 1995, 35 p. **(95/320)**

VANDENBUSSCHE H., *Is European antidumping protection against Central Europe too high ?,* June 1995, 33 p. **(95/321)**

CORTENS I. en W. NONNEMAN, *Parental education and the demand for higher education : some evidence for Belgium,* June 1995, 22 p. **(95/322)**

SCHROYEN F., *Preference characterisation and indirect Allais coefficients,* June 1995, 40 blz. **(95/323)**

DOUVEN R. en J. PLASMANS, *S.L.I.M. - A small linear interdependent model of eight EU-member states, the USA and Japan,* June 1995, 35 blz. **(95/324)**

DOUVEN R. en J. PLASMANS, *Convergence and international policy coordination in the EU : a dynamic games approach,* June 1995, 39 blz. **(95/325)**

VAN REETH D., *How the Belgian treasure should auction its debt,* July 1995, 32 blz. **(95/326)**

DUCHESNE I. en W. NONNEMAN, A neo-classical demand model for higher education in Belgium, October 1995, 52 blz. **(95/327)**

DE BORGER B. en S. WOUTERS, Transport externalities and optimal pricing and supply decisions in urban transportation : a simulation analysis for Belgium, October 1995, 52 blz. **(95/328)**